
Introduction 

The FT-817 is a very versatile rig.  However, it does benefit from more output power with 
an external PA.  It also makes a very good IF rig for a microwave transverter. 

This document describes a simple circuit based on a PICAXE that can be used to either: 

• Control one or two Power Amplifiers on different bands 

• Control one or two transverters 

Switching between the two external units is as simple as changing bands on the FT-817. 

 
Operation 

The FT-817 has two independent antenna outlets – a BNC socket on the front panel and 
an SO-239 socket at the rear.  The rig can be configured to use one or other of these 
outlets on a per-band basis.  So, two Power Amplifiers for different bands, or two 
Transverters with different IF frequencies (e.g. 2m & 70cm) can be connected 
simultaneously to the rig. 

The ACC socket on the rear panel of the FT-817 provides the signals required to control 
these external units.  The pinout of the socket is shown in Figure 1. 

 

 

Figure 1 – FT-817 ACC Socket Pinout 

The pins of interest are: 

• TX GND pulled to ground in Transmit mode (e.g. PTT pressed); 

• TX INH prevents any RF output when at a high level (>5V); and   

• BAND DATA which is an analog voltage output indicating which band is selected on 
the rig.  A table showing the output voltage for each band is given in Figure 2. 

 

 

Figure 2 – BAND DATA Voltage Levels 

 



Operation of the sequencer proceeds as follows: 

• Wait until the rig is placed into Transmit mode (indicated by TX GND pulled low); 

• Check which band is selected on the FT-817.  If it’s not 2m or 70cm, do nothing.  
Otherwise, enable the corresponding PTT output (low); 

• After a 100mS delay, take the TX INH pin low (it is at +13.8V at all other times); 

• Wait until the rig goes back to Receive mode (TX GND open); 

• Take TX INH high again; 

• Disable all PTT outputs; and 

• Loop back to wait for Transmit Mode again. 

 

Circuit 

The circuit is based around a PICAXE-08M processor.  See Figure 3. 

 

Figure 3 – Sequencer Circuit 

 

Power (+13.8V) is taken from the ACC socket.  The Programming socket (PGM) suits the 
standard programming cable available for the PICAXE. 

 

Software 

The software was written using the PICAXE Programming Editor. 



' FT-817 Sequencer 
' 
' Version 0.1  31 May 2005 
' 
' Written by David Smith VK3HZ 
' 
' Purpose: 
'  Interfaces an FT-817 to two transverters. 
'  Each transverter has a different IF (2m/70cm) and is connected to front or rear antenna port. 
'  The PICAXE activates the PTT to the appropriate transverter depending on the FT-817 band. 
'  It also sequences PTT by applying Tx_Inhibit for 100mS during switching to stop the FT-817  
'  from transmitting until all transverter switching has completed.  
'    
'       Version History: 
'  0.1 31/05/05 First attempt 
' 
'--------------------------------------------------------------------------------------------------------------- 
'                               HARDWARE  CONNECTIONS 
'                                   _____________   
'               +5V             Vcc|      U      |Gnd   0V 
'               Programming   Serin|             |OP0   FT-817 Tx Inhibit  
'               70cm IF PTT     OP4|  PICAXE08M  |IP1   FT-817 Band Select (Analog)  
'               FT-817 Tx Gnd   IP3|_____________|OP2   2m IF PTT  
' 
'PICAXE08M I/O pinouts 
'                     FUNCTION                Application in this project 
'PIN0     (pin 7)     OUT (Serial out)        FT-817 Tx Inhibit 
'PIN1     (pin 6)     IN (Analog)             FT-817 Band Select 
'PIN2     (pin 5)     OUT                     2m IF PTT 
'PIN3     (pin 4)     IN                      FT-817 Tx Gnd 
'PIN4     (pin 3)     OUT                     70cm IF PTT  
'SERIAL   (pin 2)     (Serial in)             Programming 
' 
'--------------------------------------------------------------------------------------------------------------- 
' Define constants 
' 
symbol PTT_Delay = 100    ' Delay between FT-817 Tx Ground and release of Tx Inhibit 
' 
symbol Band_2m   = 176    ' Minimum ADC reading if   2m band selected 
symbol Band_70cm = 190    ' Minimum ADC reading if 70cm band selected 
' 
' Define variables 
' 
symbol Band_Select    = b0   ' FT-817 Band Select ADC reading 
' 
'--------------------------------------------------------------------------------------------------------------- 
' Initialisation 
' 
 dirs = %00010101     ' Set Pins 0, 2, 4 as Outputs 
 pins = %00000000     ' Set all outputs low 
' 
'--------------------------------------------------------------------------------------------------------------- 
' Main Loop 
' 
' Nothing happens until FT-817 PTT is pressed 
' 
wait_tx: 
 if pin3 = 1 then wait_tx    ' Loop waiting for Tx Gnd signal from FT-817 
' 
' Check FT-817 band setting 
' 
 readadc 1, Band_Select 
 if Band_Select < Band_2m   then wait_rx  ' Is it 2 or 70?  If not, do nothing 
 if Band_Select < Band_70cm then process_2m  ' Is it 2? 
' 
' 70cm 
' 
 high 4       ' Activate 70cm PTT 
 goto process_tx 
' 
' 2m 
' 
process_2m: 
 high 2       ' Activate 2m PTT 
' 
' Hold off FT-817 transmission until switching has settled 
' 
process_tx: 
 pause PTT_Delay      ' Wait a while 
 high 0       ' Allow FT-817 to transmit 
' 
' Wait until PTT released 
' 
wait_rx: 
 if pin3 = 0 then wait_rx    ' Loop waiting for Tx Gnd signal released by FT-817 
' 
 pins = %00000000     ' Set all outputs low again 
 goto wait_tx      ' Go back and repeat it all again 
' 
 end 

 



Implementation 

If this is your first PICAXE project, then I’d recommend getting the PICAXE-08M Starter 
Pack.  This will provide you with the Programming Cable, a Prototype Board on which the 
sequencer can be built, a PICAXE-08M device and all of the software on CD. 

Otherwise, the Programming Software can be downloaded free from: 
http://www.rev-ed.co.uk/picaxe/ 

Details for building your own Programming Cable (RS-232 version) can also be found on 
that site. 

The PICAXE-08M is available from a number of suppliers. 

 

Operation 

Switch the rig to FM on 2m with PO displayed.  Press the PTT.  The 2m PA/transverter 
should immediately switch, but the power output indication on the FT-817 should be 
delayed very slightly.  On 70cm, the result should be the same.   

Now switch to 6m and try the same test.  This time, neither of the external units should 
react and there should be no output from the FT-817. 

Note that the 13.8V pin on the ACC socket is always live, even when the rig is off.  This 
means that, if the FT-817 is run on batteries, they may be run flat if the sequencer is left 
connected. 

 

Possible Modifications 

The sequencer can be changed to suit different band combinations (e.g. HF / 2m) quite 
easily by modifying the two lines in the code after the READADC line.  Approximate values 
returned by READADC for each band setting are shown in the table below: 

Band 1.8 3.5 7 10 14 18 21 24.5 28 50 144 432 

Voltage 0.33 0.67 1.00 1.33 1.67 2.00 2.33 2.67 3.00 3.33 3.67 4.00 

READADC 16 32 48 64 80 96 112 128 144 160 176 192 

 


